An Algorithm for Type-Checking Dependent Types

Thierry Coquand

Chalmers University, S 41296, Göteborg, Sweden

Abstract

We present a simple type-checker for a language with dependent types and let expressions, with a simple proof of correctness.

Introduction

Type Theory provides an interesting approach to the problem of (interactive) proof-checking. Instead of introducing, like in LCF [10], an abstract data type of theorems, it uses the proofs-as-programs analogy and reduces the problem of proof checking to the problem of type-checking in a programming language with dependent types [5]. This approach presents several advantages, well described in [11,5], among those being the possibility of independent proof verification and of a uniform treatment for naming constants and theorems. It is crucial however for this approach to proof-checking to have a simple and reliable type-checking algorithm. Since the core part of such languages, like the ones described in [5,7], seems very simple, there may be some hope for such a short and simple type-checker for dependent types. Indeed, de Bruijn sketches such an algorithm in [5]. However, this last paper leaves unspecified the treatment of conversion of terms, and more importantly, the treatment of α-conversion, and names of variables.

Though this problem of α-conversion, and the related problem of the definition of substitution, may seem at first of small importance, there are both theoretical and practical evidence that it is an important issue for languages based on λ-calculus. We can cite here Abelson and Sussman [2]: “Despite the fact that substitution is a “straightforward idea”, it turns out to be surprisingly complicated to give a rigorous mathematical definition of the substitution process . . . Indeed, there is a long history of erroneous definitions of substitution in the literature of logic and programming semantics.”

From a theoretical side, the problem of substitution and α-conversion are analysed in detail in A. Stoughton’s paper [17], and one motivation behind the

Preprint submitted to Elsevier Science 1 March 1996 calculus of explicit substitution [1] was to handle precisely these problems. One other attempt for making precise the substitution operation is the substitution calculus of P. Martin-Löf, presented in the references [19]. Unexpectedly, R. Pollack discovered that this calculus is not closed under α-conversion [16], and this illustrates well the subtlety of this topic.

From the implementation side, it is known that the first implementation of substitution in Automath [5] was incorrect, and that most of the bugs in the implementation of LCF came from clashes of bound variables in strange situations [15]. How to handle names properly is seen as one of the main problem in the implementation of a language based on Type Theory by Hanna and Daeche [11].

Despite its importance, the problem of α-conversion is relatively seldom emphasised and analysed in the literature of type-checking dependent types. Few papers are explicit on this point, and even less try to argue about the correctness of their treatment of names of variables (for some exceptions, see [11,14,16,1,18]; the references [1,16,18] are more explicit about correctness issues). When the problem is analysed in detail, like in the proof of the “substitution lemma” in Stoy’s book on denotational semantics [18], the arguments lack of conceptual content and it is difficult to grasp intuitively what makes the whole proof work.

The goal of this note is to present a simple type-checking algorithm for dependent types, with a simple proof of correctness. The main ingredient, which is the explicit introduction of closures, has been already suggested in [9], for the analysis of environment machines. In this way, we are completely explicit relatively to α-conversion, but we don’t require complicated syntactical lemmata such as the substitution lemma.

For reason of simplicity, we have chosen to illustrate this method on the simplest possible type system with dependent types, namely a type system with dependent product and as only primitive type a type Type of all types. We prove only the soundness of our type-checking algorithm here. Indeed, it is known [8], that there is no decision procedure for the typing problem if we have a type of all types. However, with only minor variations, the same algorithm can be used as the basis of a decision procedure for Martin-Löf type theory or the normalising type systems described in [4]. We give in an appendix a Gofer/Haskell implementation [3].

1 Language and Semantics of Dependent Types

1.1 Expressions

Our expression language will be the one of a type theory with dependent type and a type of all types. Let Ide be an infinite set of identifiers. The set Exp of expressions is inductively defined by:

– Ide ⊆ Exp, – Type ∈ Exp, – if M, N ∈ Exp, x ∈ Ide then λx M, M N, (x : M)N ∈ Exp. In the following, let A, B, C, M, N be expressions, x, y, z be identifiers.

For instance (A : Type)(x : A)A will be the type of the polymorphic identity function id = λAλx x. If B is a given type, id B is the identity function over the type B.

By structural induction, we can associate with any expression M its set of free variables FV(M) as usual.

1.2 Models

We don’t need to make completely precise the notion of models, but only to list some general operations and properties that any “reasonable” model should have. We use the notion of models described in [13], due to Hindley and Longo. This notion can be traced back to Henkin in the framework of simply typed λ-calculus [12].

First, we define the set Env of environments associated with a given set D and dom(ρ) ⊆ Ide, defined for ρ ∈ Env. The set Env consists of the empty environment () and of the update environment (ρ, x = d) for ρ ∈ Env and x ∈ Ide, d ∈ D. We take furthermore dom(()) = ∅ and dom((ρ, x = u)) = dom(ρ) ∪ {x}. We define lookup x ρ for x ∈ dom(ρ), by taking lookup x (ρ, y = d) to be d if x = y and lookup x ρ otherwise.

A model is a tuple (D, App, eval, Type, :) where D is a set, App : D→[D→D] a binary operation, and eval M ρ is an element of D for M expression and ρ ∈ Env such that FV(M) ⊆ dom(ρ)^1 . The meaning of Type and the relation

1 With respect to the presentation of Hindley-Seldin [13], we have made the following change: we replace environments as functions in Ide!D, by suitable finite representations of these functions.

:⊆ D × D are explained below.

First, we require that (D, App, eval) forms a model of the untyped λ-calculus, that is:

– App (eval (λx M) ρ) a = eval M (ρ, x = a), – eval x ρ = lookup x ρ, – eval (M_1 M_2) ρ = App (eval M_1 ρ) (eval M_2 ρ), – if, for all elements d ∈ D, we have eval M (ρ, x = d) = eval N (ν, y = d) then we have also eval (λx M) ρ = eval (λy N) ν.

The first condition can be seen as a semantical version of β-conversion. The last condition is called Berry’s condition in [13]. This is a quite elegant definition of model of λ-calculus, which is presented as exercise 11.9 in reference [13]. We write [[M]] for eval M ().

The situation is richer here because we have a special constant Type and a product operation. We add the conditions

– eval Type ρ = Type for all ρ, – if eval A ρ = eval C ν and eval B (ρ, x = d) = eval D (ν, y = d), for all d ∈ D, then we have also eval ((x : A)B) ρ = eval ((y : C)D) ν.

Finally, we have to express in some way that some expressions denote types, that represent collections of objects. For this, we introduce a typing relation :⊆ D × D, written infix; the relation a : d means intuitively that the value a ∈ D is of type d.

– Type : Type, – eval ((x : A)B) ρ : Type if first, eval A ρ : Type and second, a : eval A ρ implies eval B (ρ, x = a) : Type, – App c a : eval B (ρ, x = a) if c : eval ((x : A)B) ρ and a : eval A ρ, – eval (λy N) ν : eval ((x : A)B) ρ if a : eval A ρ implies eval N (ν, y = a) : eval B (ρ, x = a). The following lemma can be proved by structural induction on the expression M. Lemma 1 If FV(M) ⊆ dom(ρ), FV(M) ⊆ dom(ν), and lookup x ρ = lookup x ν for all x ∈ FV(M), then eval M ρ = eval M ν.

When later on we refer to a “model D”, we shall mean by this any structure (D, App, eval, Type, :) that satisfies all the conditions listed above.

2 The Type System

2.1 Expressions and Values

Let G be an infinite set of new variables, the generic values. We write v_1, v_2, v_3, . . . for elements of G.

Then we define by simultaneous induction the set of values V and the set of environments Env together with dom(ρ) ⊆ V for ρ ∈ Env :

– G ⊆ V, – if u, w ∈ V, then u w ∈ V, – Type ∈ V, – if M is an expression, ρ ∈ Env, FV(M) ⊆ dom(ρ), then Mρ ∈ V, – () ∈ Env, and dom(()) = ∅, – if ρ ∈ Env, x ∈ Ide, u ∈ V, then (ρ, x = u) ∈ Env, and dom((ρ, x = u)) = dom(ρ) ∪ {x}. We define lookup x ρ as before. In the following, assume u, v, w ∈ V.

Any assignment f ∈ G→D extends uniquely to an f ∈ V→D such that

– f Type = Type, – f (u_1 u_2) = App (f u_1) (f u_2), – f (Mρ) = eval M (f^�ρ),

where f^�() = () and f^�(ρ, x = u) = ((f^�ρ), x = f u).

We can define inductively when a generic value occurs in a given value, and prove that f u = g u if f and g agree on all generic values that occur in u.

2.2 Conversion Relation

Conversion applies only to values. Like for λ-calculus, we use the ordinary equality symbol u = v to denote the fact that the values u and v are convertible. This relation is defined inductively as the least congruence such that (notice that this congruence appears only positively in the clauses that follow, and this is why we can define conversion inductively): – (λx M)ρ v = M(ρ, x = v), – x(ρ, x = u) = u, – x(ρ, y = v) = xρ, if x ≠ y, – (M N)ρ = (Mρ) (Nρ),

– if v_k does not occur in ρ, ν, and M(ρ, x = v_k) = N(ν, y = v_k), then (λx M)ρ = (λy N)ν, – Type ρ = Type, – if v_k does not occur in ρ, ν, and Aρ = Bν, B(ρ, x = v_k) = D(ν, y = v_k), then ((x : A)B)ρ = ((y : C)D)ν.

The following Lemma is similar to Lemma 1, and follows from the fact that the set G of generic values is infinite.

Lemma 2 If FV(M) ⊆ dom(ρ), FV(M) ⊆ dom(ν), and lookup x ρ = lookup x ν for all x ∈ FV(M), then Mρ = Mν.

The following property expresses the soundness of our notion of conversion between values.

Proposition 1 If u_1 and u_2 are convertible value, then for any model D, and any assignment f ∈ G→D, we have f u_1 = f u_2 in D.

Proof: We present the rule corresponding to Berry’s condition, because this is the only delicate case.

We have to show eval (λx M) (f^�ρ) = eval (λy N) (f^� ν) in D, given that M(ρ, x = v_k) = N(ν, y = v_k) where v_k does not occur in ρ, ν. Let d be an arbitrary element of D. Let then g be the assignment that differs from f only on the variable v_k and such that g v_k = d. Because v_k does not occur in ρ, ν we have f^�ρ = g^�ρ and f^�ν = g^�ν, and hence ((f^� ρ), x = d) = g^� (ρ, x = v_k) and ((f^� ν), x = d) = g^� (ν, x = v_k). So, by induction, eval M ((f^� ρ), x = d) = eval N ((f^� ν), y = d). Since this holds for all d ∈ D, we get eval (λx M) (f^�ρ) = eval (λy N) (f^� ν) because Berry’s condition holds for D.

2.3 Conversion algorithm

2.3.1 Weak Head Normal Form

The conversion algorithm uses an algorithm that computes the weak head normal form of a value. This algorithm is represented by a relation u ⇓ u^0 between values, which can be read as u evaluates to u^0. This relation is inductively defined by: – if u ⇓ (λx M)ρ, and M(ρ, x = w) ⇓ v, then u w ⇓ v, – if u ⇓ u^0 and u^0 is a generic value or an application, then u w ⇓ u^0 w, – if lookup x ρ ⇓ v, then xρ ⇓ v, – Type ρ ⇓ Type,

– if (Mρ) (Nρ) ⇓ v, then (M N)ρ ⇓ v, – v ⇓ v if v is of the form v_k or Mρ, where M is an abstraction or a product.

Notice that this relation is partial and deterministic. Furthermore,

Lemma 3 If u ⇓ v, then u = v.

Corollary 1 If u ⇓ v, then f u = f v ∈ D, for any model D, and any assignment f ∈ G→D.

2.3.2 Conversion

The conversion algorithm is represented by a relation u_1 ∼ u_2. We define inductively that u_1 ∼ u_2 holds if:

– u_1 ⇓ Type and u_2 ⇓ Type, or – u_1 ⇓ t_1 w_1, u_2 ⇓ t_2 w_2, and t_1 ∼ t_2 and w_1 ∼ w_2, or – u_1 ⇓ v_{k1} , u_2 ⇓ v_{k2} and k_1 = k_2, or – u_1 ⇓ (λx_1 M_1)ρ_1, u_2 ⇓ (λx_2 M_2)ρ_2 and M_1(ρ_1, x_1 = v_k) ∼ M_2(ρ_2, x_2 = v_k) where v_k does not occur in ρ_1, ρ_2, or – u_2 ⇓ ((x_1 : A_1)B_1)ρ_1, u_2 ⇓ ((x_2 : A_2)B_2)ρ_2 and we have both A_1ρ_1 ∼ A_2ρ_2 and B_1(ρ_1, x_1 = v_k) ∼ B_2(ρ_2, x_2 = v_k) where v_k does not occur in ρ_1, ρ_2.

From Lemma 3, we get the semantical soundness of this algorithm.

Lemma 4 If u_1 ∼ u_2, then u_1 = u_2.

Corollary 2 If u_1 ∼ u_2, then f u_1 = f u_2 for any model D and any assignment f ∈ G→D.

2.4 Type-Checking Algorithm

Ultimately, we want to check when an expression A is a correct type, and given such an expression A, when an expression M is of type A. As a technical intermediary notion, it is convenient to introduce a typing relation between expressions and values: M is of type u will mean that the value M() is of type u. Recursively, we need to express when a given expression M, in a given pair of environments ρ and Γ, is of type u, where u is a value. Intuitively, ρ assigns values and Γ type values to the free variables of M. We write this relation ρ; Γ ⊢ M ⇒ v. We need to define simultaneously the type inference relation ρ; Γ ⊢ M → v, meaning that it is possible to infer the type value v for M in the environments ρ and Γ.

The type-checking algorithm ρ; Γ ⊢ M ⇒ v and the type inference algorithm ρ; Γ ⊢ M → v are represented by two relations defined inductively and simultaneously.

– If v ⇓ ((y : A)B)ρ^0, and ρ, x = v_k; Γ, x : Aρ^0 ⊢ N ⇒ B(ρ^0, y = v_k) where v_k does not occur in ρ, Γ, ρ^0, then ρ; Γ ⊢ λy N ⇒ v, – if ρ; Γ ⊢ A ⇒ Type, and ρ, x = v_k; Γ, x : Aρ ⊢ B ⇒ Type, where v_k does not occur in ρ, Γ, and v ⇓ Type, then ρ; Γ ⊢ (x : A)B ⇒ v, – if ρ; Γ ⊢ M → w and w ∼ v, then ρ; Γ ⊢ M ⇒ v, – ρ; Γ ⊢ x → xΓ, if x occurs in Γ, – if ρ; Γ ⊢ M_1 → u_1 and u_1 ⇓ ((x : A)B)ρ^0, and ρ; Γ ⊢ M_2 ⇒ Aρ^0, then ρ; Γ ⊢ M_1 M_2 → B(ρ^0, x = M_2ρ), – ρ; Γ ⊢ Type → Type.

Proposition 2 If ⊢ A ⇒ Type and ⊢ M ⇒ A(), then [[A]] : Type and [[M]] : [[A]] in D.

Proof: We prove more generally, by simultaneous induction on the definition of the two relations of type-checking and type inference, that, for any assignment f such that eval x (f^� ρ) is of type eval x (f^� Γ) in D for all x ∈ dom(Γ), if ρ; Γ ⊢ M ⇒ u or ρ; Γ ⊢ M → u, then eval M (f^� ρ) is of type f u in D.

We illustrate only the abstraction case of the type-checking relation, more delicate than the other cases. We have to check that, for any suitable assignment f, the value f ((λx N)ρ) is of type f v, with f v = eval (y : A)B (f^� ρ^0). For this it is enough to check that eval N ((f^� ρ), x = d) is of type eval B ((f^� ρ^0), y = d) for any d ∈ D which is of type eval A (f^� ρ^0).

Let g be the assignment that differs from f only on the variable v_k and such that g v_k = d. Because v_k does not occur in ρ, ρ^0, we have eval N ((f^� ρ), x = d) = eval N (g^� (ρ, x = v_k)) and eval B ((f^� ρ^0), y = d) = eval B (g^� (ρ^0, y = v_k)), and eval A (f^�ρ^0) = eval A (g^� ρ^0). Hence the result follows by induction hypothesis.

Notice that our algorithm accepts the following judgement ⊢ λxλx x : (x : A)(y : P x)P x, which is not derivable in Martin-Löf’s substitution calculus [16].

3 Extension to let expressions

This treatment extends directly to the addition of let expressions. We add expressions of the form let x = M : A in N. The meaning of this expression is reflected by the conversion rule

(let x = M : A in N)ρ = N(ρ, x = Mρ).

The typing rule is that ρ; Γ ⊢ let x = M : A in N ⇒ v iff ρ; Γ ⊢ A ⇒ Type and ρ; Γ ⊢ M ⇒ Aρ and ρ, x = Mρ; Γ, x : Aρ ⊢ N ⇒ v. We get a language that is similar to de Bruijn’s ΛΔ [5]. The problem of type-checking such let expressions is explained and motivated with a concrete example at the end of the survey article [4].

Related Works and Conclusion

We have presented a simple implementation and correctness proof of a typechecking algorithm for dependent types, while being explicit about the problem of α-conversion. This is made possible by the explicit introduction of closures.

Previous attempts of a complete description of a type-checking algorithm for dependent types can be found in [14,16,5]. In reference [1] there is presented a complete type-checking algorithm for second-order lambda-calculus, that contains most of the difficulty of type-checking dependent types. This algorithm has been used in the language Quest [6]. As can be seen by comparing this algorithm with the algorithm we present, our approach is more straightforward. A closer formalism is a predecessor of this work on explicit substitution, presented in [9], which introduces the idea of explicit closures.

We think that the same method can be used to simplify the presentation of the semantics of languages with a binding structure, and the meta-mathematical analysis of languages with dependent types [4].

Acknowledgement

This work owes much to several discussions with Dan Synek. Randy Pollack and Luca Cardelli made valuable comments on a preliminary version. Thanks to Bernhard Möller and to the referee for their comments concerning the presentation of this note.

References

[1] M.Abadi, L.Cardelli, P.L. Curien, J.J. Levy. Explicit Substitutions. Journal of Functional Programming 1 (4), p. 375-416, 1991.

[2] H. Abelson and G. Sussman. Structure and Interpretation of Computer Programs. The MIT Press, 1986.

[3] L. Augustsson. Haskell B. user’s manual available over WWW from hhtp://www.cs.chalmers.se:80/pub/haskell/chalmers

[4] H. Barendregt. Lambda Calculi with Types. Handbook of Logic in Computer Science, Volume II, Edited by S. Abramski, D.M. Gabbai and T.S.E. Maibaum, Oxford University Press.

[5] N.G. de Bruijn. A plea for weaker frameworks. in Logical Framework, G. Huet and G. Plotkin eds, Cambridge University Press, p. 40 - 68, 1991.

[6] L. Cardelli. Typeful Programming. in: Formal Description of Programming Concepts, E.J. Neuhold,.Paul Eds., Springer-Verlag, 1991.

[7] Th. Coquand and G. Huet. The Calculus of Constructions. Information and Computation 76, p. 95 - 120, 1988.

[8] Th. Coquand and H. Herbelin. A-translation and looping combinators in pure type system. Journal of Functional Programming, 1994, vol 4, p. 77-88.

[9] P.L. Curien. An abstract framework for environment machines. Theoretical Computer Science 82, p. 389-402, 1991.

[10] M.J. Gordon, A.J. Milner and C.P. Wadsworth. Edinburgh LCF - a Mechanised Logic of Computation. Vol. 78, Lecture Notes in Computer Science, Springer- Verlag, 1979.

[11] K. Hanna, N. Daeche. Purely Functional Implementation of a Logic. Vol. 230, Lecture Notes in Computer Science, Springer-Verlag, p. 598-607, 1986.

[12] L. Henkin. Completeness in the theory of types. Journal of Symbolic Logic, 15, p. 81-91, 1950.

[13] J.R. Hindley and J. Seldin. Introduction to Combinators and ˘-calculus.

London Mathematical Society Student Texts 1, Cambridge University press, 1986.

[14] G. Huet. The Constructive Engine. In R. Narasimhan, editor, A Perspective in Theoretical Computer Science, World Scientific Publishing, 1989.

[15] L. Paulson. Isabelle: The Next 700 Theorem Provers. In: P. Odifreddi, ed., Logic and Computer Science, 1990, p. 361-386.

[16] R. Pollack. Closure under Alpha Conversion. In: Types for Proofs and Programs, H. Barendregt, T. Nipkow eds., Vol. 806, Lecture Notes in Computer Science, Springer-Verlag, p. 313 - 332, 1993.

[17] A. Stoughton. Substitution Revisited. Theoretical Computer Science 59 (1988), p. 317-325.

[18] J. Stoy. Denotational Semantics. MIT Press, Cambridge, 1977 [19] A. Tatsitro. Formulation of Martin-Löf’s Theory of Types with Explicit Substitution. Licentiate Thesis, Chalmers University, 1993.

[20] G. Winskel. The formal semantics of prorgamming languages, an introduction.

The MIT Press, 1993

Appendix: a Gofer/Haskell implementation

-- the main data types and general functions

type Id = String

data Exp =
 Var Id | App Exp Exp | Abs Id Exp | Let Id Exp Exp Exp
 | Pi Id Exp Exp | Type

data Val = VGen Int | VApp Val Val | VType | VClos Env Exp

type Env = [(Id,Val)]

update :: Env -> Id -> Val -> Env
update env x u = (x,u):env

lookup :: Id -> Env -> Val
lookup x ((y,u):env) =
 if x == y then u
 else lookup x env
lookup x [] = error ("lookup " ++ x)

-- a short way of writing the whnf algorithm

app :: Val -> Val -> Val
eval :: Env -> Exp -> Val

app u v =
 case u of
   VClos env (Abs x e) -> eval (update env x v) e
   _ -> VApp u v

eval env e =
 case e of
   Var x -> lookup x env
   App e1 e2 -> app (eval env e1) (eval env e2)
   Let x e1 _ e3 -> eval (update env x (eval env e1)) e3
   Type -> VType
   _ -> VClos env e

whnf :: Val -> Val
whnf v =
 case v of
   VApp u w -> app (whnf u) (whnf v)
   VClos env e -> eval env e
   _ -> v

-- the conversion algorithm; the integer is
-- used to represent the introduction of a fresh variable

eqVal :: (Int,Val,Val) -> Bool
eqVal (k,u1,u2) =
 case (whnf u1,whnf u2) of
   (VType,VType) -> True
   (VApp t1 w1,VApp t2 w2) ->
      eqVal (k,t1,t2) && eqVal (k,w1,w2)
   (VGen k1,VGen k2) -> k1 == k2
   (VClos env1 (Abs x1 e1),VClos env2 (Abs x2 e2)) ->
      let v = VGen k
      in eqVal (k+1,
                VClos (update env1 x1 v) e1,
                VClos (update env2 x2 v) e2)
   (VClos env1 (Pi x1 a1 b1),VClos env2 (Pi x2 a2 b2)) ->
      let v = VGen k

      in eqVal (k,VClos env1 a1,VClos env2 a2) &&
         eqVal (k+1,
                VClos (update env1 x1 v) b1,
                VClos (update env2 x2 v) b2)
   _ -> False

-- type-checking and type inference

checkExp :: (Int,Env,Env) -> Exp -> Val -> Bool
inferExp :: (Int,Env,Env) -> Exp -> Val
checkType :: (Int,Env,Env) -> Exp -> Bool

checkType (k,rho,gamma) e = checkExp (k,rho,gamma) e VType

checkExp (k,rho,gamma) e v =
 case e of
   Abs x n ->
    case whnf v of
      VClos env (Pi y a b) ->
        let v = VGen k
        in checkExp (k+1,
                     update rho x v,
                     update gamma x (VClos env a))
                    n (VClos (update env y v) b)
      _ -> error"expected Pi"
   Pi x a b ->
     case whnf v of
       VType -> checkType (k,rho,gamma) a &&
               checkType (k+1,
                          update rho x (VGen k),
                          update gamma x (VClos rho a))
                        b
       _ -> error"expected Type"
   Let x e1 e2 e3 ->
     checkType (k,rho,gamma) e2 &&
     checkExp (k,
               update rho x (eval rho e1),
               update gamma x (eval rho e2))
              e3 v
   _ -> eqVal (k,inferExp (k,rho,gamma) e,v)

inferExp (k,rho,gamma) e =
 case e of
   Var id -> lookup id gamma
   App e1 e2 ->
      case whnf (inferExp (k,rho,gamma) e1) of
        VClos env (Pi x a b) ->
          if checkExp (k,rho,gamma) e2 (VClos env a)
               then VClos (update env x (VClos rho e2)) b
          else error"application error"
        _ -> error"application, expected Pi"
   Type -> VType
   _ -> error"cannot infer type"

typecheck :: Exp -> Exp -> Bool

typecheck m a =
 checkType (0,[],[]) a &&
 checkExp (0,[],[]) m (VClos [] a)

test :: Bool
test =
 typecheck (Abs "A" (Abs "x" (Var "x")))
           (Pi "A" Type (Pi "x" (Var "A") (Var "A")))